



High-Contrast Imaging via Pupil Masking

Robert J. Vanderbei

July 24, 2013

AFTA-Math Workshop
PCTS
Princeton Univ.


<http://www.princeton.edu/~rvdb>

High-Contrast Imaging for Planet-Finding

Build a telescope capable of finding Earth-like planets around nearby Sun-like stars.

Problem is hard:

- Star is 10^{10} times brighter than the planet.
- Angular separation is small ≈ 0.05 arcseconds.
- Light is a wave: the star is not a pinpoint of light—it has a *diffraction pattern*.
- Light is particles (photons) \Rightarrow Poisson statistics.



The diffraction pattern is the magnitude-squared of the *Fourier transform* of the telescope's *pupil*.

Pupil Masking/Apodization



Pupil Masking/Apodization



Fourier/Fresnel Propagation

We start with the Fresnel propagation formula for propagating an electric field a distance f after a lens/mirror of focal length f :

$$E_{\text{out}}(\xi, \eta) = \frac{i}{\lambda f} e^{\frac{\pi i}{\lambda f}(\xi^2 + \eta^2)} \iint e^{-\frac{2\pi i}{\lambda f}(x\xi + y\eta)} A(x, y) E_{\text{in}}(x, y) dy dx.$$

Here,

- E_{in} denotes the *electromagnetic field* at the *pupil*;
- E_{out} denotes the *electromagnetic field* at the *image*;
- A denotes the *apodization*;
- Temporal variations are suppressed.

At the design stage, we can

- Assume that the input field E_{in} is an on-axis plane wave;
- Normalize so that $E_{\text{in}} \equiv 1$;
- Choose units to be *aperture*'s in the pupil plane and λ/D in the image plane;
- Drop leading unit magnitude factors.

Result: a simple *2D Fourier transform*.

Fourier Propagation in Simple Units

Let $f(x, y)$ denote the transmissivity (i.e., *apodization*) at location (x, y) on the surface of a filter placed over the pupil of a telescope.

The *electromagnetic field* in the image plane of such a telescope associated with an on-axis point source (i.e., a star) is proportional to the 2D Fourier transform of the apodization f :

$$\hat{f}(\xi, \eta) = \iint e^{2\pi i(x\xi + y\eta)} f(x, y) dx dy.$$

The *intensity* of the light in the image is proportional to the *magnitude squared* of \hat{f} .

Assuming that the underlying telescope has a circular opening of radius one half, we impose the following constraint on f :

$$f(x, y) = 0 \quad \text{for} \quad x^2 + y^2 > 1/2^2.$$

Optimized Apodizations

Maximize *light throughput* subject to constraint that almost no light reaches a given *dark zone* \mathcal{D} and other *structural* constraints:

$$\begin{aligned} & \text{maximize} && \int_{\square} f(x, y) dx dy && \left(= \hat{f}(0, 0) \right) \\ & \text{subject to} && \left| \hat{f}(\xi, \eta) \right| \leq \varepsilon \hat{f}(0, 0), && (\xi, \eta) \in \mathcal{D}, \\ & && f(x, y) = 0, && x^2 + y^2 > 1/2^2, \\ & && 0 \leq f(x, y) \leq 1, && \text{for all } x, y. \end{aligned}$$

Here, ε is a small positive constant (on the order of 10^{-5}).

In general, the Fourier transform \hat{f} is complex valued.

This optimization problem has a *linear objective* function and both *linear* and *second-order cone* constraints.

Hence, a discretized version can be solved (to a *global optimum*).

Exploiting Symmetry

Assuming that the filter can be symmetric with respect to reflection about both axes (note: sometimes not possible), the Fourier transform can be written as

$$\hat{f}(\xi, \eta) = 4 \int_0^{1/2} \int_0^{1/2} \cos(2\pi x\xi) \cos(2\pi y\eta) f(x, y) dx dy.$$

In this case, the Fourier transform is real and so the second-order cone constraints can be replaced with a pair of inequalities,

$$-\varepsilon \hat{f}(0, 0) \leq \hat{f}(\xi, \eta) \leq \varepsilon \hat{f}(0, 0),$$

making the problem an *infinite dimensional linear programming problem*.

Note: *Curse of Dimensionality*

Exploiting Symmetry

Assuming that the filter can be symmetric with respect to reflection about both axes (note: sometimes not possible), the Fourier transform can be written as

$$\hat{f}(\xi, \eta) = 4 \int_0^{1/2} \int_0^{1/2} \cos(2\pi x\xi) \cos(2\pi y\eta) f(x, y) dx dy.$$










In this case, the Fourier transform is real and so the second-order cone constraints can be replaced with a pair of inequalities,

$$-\varepsilon \hat{f}(0, 0) \leq \hat{f}(\xi, \eta) \leq \varepsilon \hat{f}(0, 0),$$

making the problem an *infinite dimensional linear programming problem*.

Note: *Curse of Dimensionality*: $2 > 1$.

Potpourri of Pupil Masks



Discretization

Consider a two-dimensional Fourier transform

$$\widehat{f}(\xi, \eta) = 4 \int_0^{1/2} \int_0^{1/2} \cos(2\pi x\xi) \cos(2\pi y\eta) f(x, y) dx dy.$$

Its discrete approximation can be computed as

$$\widehat{f}_{j_1, j_2} = 4 \sum_{k_2=1}^n \sum_{k_1=1}^n \cos(2\pi x_{k_1} \xi_{j_1}) \cos(2\pi y_{k_2} \eta_{j_2}) f_{k_1, k_2} \Delta x \Delta y, \quad 1 \leq j_1, j_2 \leq m,$$

where

$$x_k = (k - 1/2)\Delta x, \quad 1 \leq k \leq n,$$

$$y_k = (k - 1/2)\Delta y, \quad 1 \leq k \leq n,$$

$$\xi_j = (j - 1/2)\Delta \xi, \quad 1 \leq j \leq m,$$

$$\eta_j = (j - 1/2)\Delta \eta, \quad 1 \leq j \leq m,$$

$$f_{k_1, k_2} = f(x_{k_1}, y_{k_2}), \quad 1 \leq k_1, k_2 \leq n,$$

$$\widehat{f}_{j_1, j_2} \approx \widehat{f}(\xi_{j_1}, \eta_{j_2}), \quad 1 \leq j_1, j_2 \leq m.$$

Complexity: $m^2 n^2$.

A Clever (and Trivial!) Idea

The obvious brute force calculation requires m^2n^2 operations.

However, we can “factor” the double sum into a nested pair of sums.

Introducing new variables that represent the inner sum, we get:

$$g_{j_1, k_2} = 2 \sum_{k_1=1}^n \cos(2\pi x_{k_1} \xi_{j_1}) f_{k_1, k_2} \Delta x, \quad 1 \leq j_1 \leq m, \quad 1 \leq k_2 \leq n,$$

$$\hat{f}_{j_1, j_2} = 2 \sum_{k_2=1}^n \cos(2\pi y_{k_2} \eta_{j_2}) g_{j_1, k_2} \Delta y, \quad 1 \leq j_1, j_2 \leq m,$$

Formulated this way, the calculation requires only $mn^2 + m^2n$ operations.

Brute Force vs Clever Approach

On the following page we show two AMPL model formulations of this problem.

On the left is the version expressed in the straightforward one-step manner.

On the right is the AMPL model for the same problem but with the Fourier transform expressed as a pair of transforms—the so-called *two-step process*.

The dark zone \mathcal{D} is a pair of sectors of an annulus with inner radius 4 and outer radius 20.

Except for the resolution, the two models produce the same result.

Two AMPL Models

```
param rho0 := 4;      param rho1 := 20;
param m := 35;      # discretization parameter
param n := 150;     # discretization parameter
param dx := 1/(2*n);  param dy := dx;

set Xs := setof {j in 0.5..n-0.5 by 1} j/(2*n);
set Ys := Xs;
set Pupil :=
    setof {x in Xs, y in Ys: x^2+y^2<0.25} (x,y);
set Xis := setof {j in 0..m} j*rho1/m;
set Etas := Xis;
set DarkHole := setof {xi in Xis, eta in Etas:
    xi^2+eta^2>=rho0^2 &&
    xi^2+eta^2<=rho1^2 &&
    eta <= xi } (xi,eta);

var f {(x,y) in Pupil} >= 0, <= 1;

var fhat {xi in Xis, eta in Etas};

maximize area: sum {(x,y) in Pupil} f[x,y]*dx*dy;

subject to fhat_def {xi in Xis, eta in Etas}:
    fhat[xi,eta] = 4*sum {(x,y) in Pupil}
        f[x,y]*cos(2*pi*x*xi)
            *cos(2*pi*y*eta)*dx*dy;
subject to sidelobe_pos {(xi,eta) in DarkHole}:
    fhat[xi,eta] <= 10^(-5)*fhat[0,0];
subject to sidelobe_neg {(xi,eta) in DarkHole}:
    -10^(-5)*fhat[0,0] <= fhat[xi,eta];

solve;
```

```
param rho0 := 4;      param rho1 := 20;
param m := 35;      # discretization parameter
param n := 1000;    # discretization parameter
param dx := 1/(2*n);  param dy := dx;

set Xs := setof {j in 0.5..n-0.5 by 1} j/(2*n);
set Ys := Xs;
set Pupil :=
    setof {x in Xs, y in Ys: x^2+y^2 < 0.25} (x,y);
set Xis := setof {j in 0..m} j*rho1/m;
set Etas := Xis;
set DarkHole := setof {xi in Xis, eta in Etas:
    xi^2+eta^2>=rho0^2 &&
    xi^2+eta^2<=rho1^2 &&
    eta <= xi } (xi,eta);


var f {(x,y) in Pupil} >= 0, <= 1;
var g {xi in Xis, y in Ys};
var fhat {xi in Xis, eta in Etas};

maximize area: sum {(x,y) in Pupil} f[x,y]*dx*dy;

subject to g_def {xi in Xis, y in Ys}:
    g[xi,y] = 2*sum {x in Xs: (x,y) in Pupil}
        f[x,y]*cos(2*pi*x*xi)*dx;
subject to fhat_def {xi in Xis, eta in Etas}:
    fhat[xi,eta] = 2*sum {y in Ys}
        g[xi,y]*cos(2*pi*y*eta)*dy;
subject to sidelobe_pos {(xi,eta) in DarkHole}:
    fhat[xi,eta] <= 10^(-5)*fhat[0,0];
subject to sidelobe_neg {(xi,eta) in DarkHole}:
    -10^(-5)*fhat[0,0] <= fhat[xi,eta];

solve;
```

Optimal Solution



Left. The optimal apodization found by either of the models shown on previous slide.

Center. Plot of the star's image (using a linear stretch).

Right. Logarithmic plot of the star's image (black = 10^{-10}).

Notes:

- The “apodization” turns out to be purely opaque and transparent (i.e., a mask).
- The mask has “islands” and therefore must be laid on glass.

Close Up

Brute force with $n = 150$



Two-step with $n = 1000$



Summary Problem Stats

Comparison between a few sizes of the one-step and two-step models.



Problem-specific stats.

Model	n	m	constraints	variables	nonzeros	arith. ops.
One step	150	35	976	17,672	17,247,872	17,196,541,336
One step	250	35	*	*	*	*
Two step	150	35	7,672	24,368	839,240	3,972,909,664
Two step	500	35	20,272	215,660	7,738,352	11,854,305,444
Two step	1000	35	38,272	822,715	29,610,332	23,532,807,719




Hardware/Solution-specific performance comparison data.

Model	n	m	iterations	primal objective	dual objective	cpu time (sec)
One step	150	35	54	0.05374227247	0.05374228041	1380
One step	250	35	*	*	*	*
Two step	150	35	185	0.05374233071	0.05374236091	1064
Two step	500	35	187	0.05395622255	0.05395623990	4922
Two step	1000	35	444	0.05394366337	0.05394369256	26060




Concentric Ring Masks






Circular Unobstructed Pupil






Circular Unobstructed Pupil






Circular Unobstructed Pupil






SPICA




Subaru



JWST






Throughput: $\sim 45\%$
WA: 5 to $15 \lambda/D$
Contrast: 10^{-5}



AFTA Telescope

- $iwa = 3.5 \lambda/D$, $owa = 20 \lambda/D$
- $contrast = 10^{-9}$
- $throughput \approx 10\%$
- $central\ lobe \approx 31\%$, $trilobe \approx 76\%$



Thank You!